This is an online Appendix to the paper "On regulation and competition: pros and cons of a diversified monopolist" by G. Calzolari and C. Scarpa

Model Setup: Cournot competition with differentiated products

```
(*Consumer preferences:
  using symmetry where we explicitate y1 and all other (n-1) firms' output*)
u = μ (y1 + (n - 1) y) - 1 / 2 (y1^2 + (n - 1) y^2) - γ ((n - 1) y1 y + y y (n - 2) (n - 1));

(*The restrictions on the parameters are: γ belongs to [-1,1],
negative are complements,
positive are substitutes, =0 independent and =1 perfect substitutes*)

p1 = μ - y1 - γ (n - 1) y;
p = μ - y - γ ((n - 2) yj + y1);

(*Defining profits*)
Profit1 = y1 p1 - c y1 + θ q y1;
Profit = y p - c y;
```

Analysis of competition

```
(*COMPETITION WITH INCOMPLETE
 INFORMATION: regulation is uniform and does not transmit information to the rivals*)

Profit1uAI = Profit1 /. {θ -> θu , y1 → y1u};
Profit1dAI = Profit1 /. {θ -> θd , y1 → y1d};

ProfituAI = Profit /. y1 → y1u;
ProfitdAI = Profit /. y1 → y1d;
EProfitAI = v ProfituAI + (1 - v) ProfitdAI;
```

```
DEProfitAI = Simplify[D[EProfitAI, y] /. yj → y];
FullSimplify[
 Solve[{DEProfitAI == 0, D[Profit1uAI, y1u] == 0, D[Profit1dAI, y1d] == 0}, {y, y1u, y1d}]]
```

$$\left\{\left\{y1u \to \frac{-2\,c\,(-2+\gamma) + q\left((-1+n)\,(-1+v)\,\gamma^2\,(\theta d - \theta u) - 4\,\theta u - 2\,(-2+n)\,\gamma\,\theta u\right) + 2\,(-2+\gamma)\,\mu}{2\,(-2+\gamma)\,(2+(-1+n)\,\gamma)},\right.\right.$$
$$y \to \frac{-c\,(-2+\gamma) + q\,\gamma\,(\theta d - v\,\theta d + v\,\theta u) + (-2+\gamma)\,\mu}{(-2+\gamma)\,(2+(-1+n)\,\gamma)},$$
$$\left.\left.y1d \to \frac{-2\,c\,(-2+\gamma) + q\left(\left(-4 - 2\,(-2+n)\,\gamma + (-1+n)\,v\,\gamma^2\right)\theta d - (-1+n)\,v\,\gamma^2\,\theta u\right) + 2\,(-2+\gamma)\,\mu}{2\,(-2+\gamma)\,(2+(-1+n)\,\gamma)}\right\}\right\}$$

```
(*COMPETITION WITH FULL
 INFORMATION: information has been transmitted by discriminatory regulation*)
DProfit = D[Profit, y] /. yj → y;
stmp = Simplify[Solve[{D[Profit1, y1] == 0, DProfit == 0}, {y1, y}]]
y1FI = y1 /. stmp[[1, 1]];
yFI = y /. stmp[[1, 2]];

p1FI = FullSimplify[p1 /. {y1 → y1FI, y → yFI}];
pFI = FullSimplify[p /. yj → y /. {y1 → y1FI, y → yFI}];

Profit1FI = FullSimplify[Profit1 /. {y1 → y1FI, y → yFI}]
ProfitFI = FullSimplify[Profit /. yj → y /. {y1 → y1FI, y → yFI}]
```

$$\left\{\left\{y1 \to \frac{-c\,(-2+\gamma) - q\,(2+(-2+n)\,\gamma)\,\theta + (-2+\gamma)\,\mu}{(-2+\gamma)\,(2+(-1+n)\,\gamma)},\ y \to \frac{-c\,(-2+\gamma) + q\,\gamma\,\theta + (-2+\gamma)\,\mu}{(-2+\gamma)\,(2+(-1+n)\,\gamma)}\right\}\right\}$$

$$\frac{\left(c\,(-2+\gamma) + q\,(2+(-2+n)\,\gamma)\,\theta - (-2+\gamma)\,\mu\right)^2}{(-2+\gamma)^2\,(2+(-1+n)\,\gamma)^2}$$

$$\frac{\left(-c\,(-2+\gamma) + q\,\gamma\,\theta + (-2+\gamma)\,\mu\right)^2}{(-2+\gamma)^2\,(2+(-1+n)\,\gamma)^2}$$

```
(*With screening regulation, off the equilibrium the conglomerate reports untruthfully
 to the regulator and hence the information passed to the rivals is manipulated:
  we use the notation that the first letter refers to the
   true type and the second to the announced one, i.e. y1ud is the
 quantity of the conglomerate when its true type is θu but it announces θd
*)

(*Low type when announcing truthfully*)
y1dd = y1FI /. {θ -> θd , q → qd};
ydd = yFI /. {θ -> θd , q → qd};
Profit1dd = FullSimplify[Profit1 /. {y1 → y1dd, y → ydd} /. {θ -> θd , q → qd}]
Profitdd = FullSimplify[Profit /. yj → y /. {y1 → y1dd, y → ydd} /. {θ -> θd , q → qd}]
```

$$\frac{\left(c\,(-2+\gamma) + qd\,(2+(-2+n)\,\gamma)\,\theta d - (-2+\gamma)\,\mu\right)^2}{(-2+\gamma)^2\,(2+(-1+n)\,\gamma)^2}$$

$$\frac{\left(-c\,(-2+\gamma) + qd\,\gamma\,\theta d + (-2+\gamma)\,\mu\right)^2}{(-2+\gamma)^2\,(2+(-1+n)\,\gamma)^2}$$

```
(*High type when announcing truthfully*)
y1uu = y1FI /. {θ -> θu , q → qu};
yuu = yFI /. {θ -> θu , q → qu};
Profit1uu = FullSimplify[Profit1 /. {y1 → y1uu, y → yuu} /. {θ -> θu , q → qu}]
Profituu = FullSimplify[Profit /. yj → y /. {y1 → y1uu, y → yuu} /. {θ -> θu , q → qu}]
```

$$\frac{(c\,(-2+\gamma)+qu\,(2+(-2+n)\,\gamma)\,\theta u-(-2+\gamma)\,\mu)^2}{(-2+\gamma)^2\,(2+(-1+n)\,\gamma)^2}$$

$$\frac{(-c\,(-2+\gamma)+qu\,\gamma\,\theta u+(-2+\gamma)\,\mu)^2}{(-2+\gamma)^2\,(2+(-1+n)\,\gamma)^2}$$

```
(*High type when announcing untruthfully*)
y1ud =
  y1 /. FullSimplify[Solve[{0 == D[Profit1, y1] /. y → ydd}, {y1}] /. {q → qd, θ → θu}][[1, 1]];
yud = ydd;
Profit1ud = FullSimplify[Profit1 /. {y1 → y1ud, y → yud} /. {θ -> θu , q → qd} ];
Profitud = FullSimplify[Profit /. yj → y /. {y1 → y1ud, y → yud} /. {θ -> θu , q → qd}] ;
```

```
FullSimplify[y1uu - y1ud]
FullSimplify[y1ud - (y1uu + ((-2 (qu - qd) (2 + (-2 + n) γ) θu + qd (-1 + n) γ^2 (θd - θu)) / (2 (2 - γ) (2 + (-1 + n) γ))))]
(*This last is a check of the rewriting used in the paper*)
```

$$\frac{-2\,qu\,(2+(-2+n)\,\gamma)\,\theta u+qd\,\left((-1+n)\,\gamma^2\,(\theta d-\theta u)+4\,\theta u+2\,(-2+n)\,\gamma\,\theta u\right)}{2\,(-2+\gamma)\,(2+(-1+n)\,\gamma)}$$

0

```
(*This is a check of the rewriting used in the paper*)
FullSimplify[yuu - yud]
FullSimplify[yud - (yuu - ((γ (qd (θd - θu) - (qu - qd) θu)) / ((2 - γ) (2 + (-1 + n) γ))))]
FullSimplify[yud - (yuu + ((-qd γ θd + qu γ θu) / ((2 - γ) (2 + (-1 + n) γ))))]
```

$$\frac{-qd\,\gamma\,\theta d+qu\,\gamma\,\theta u}{(-2+\gamma)\,(2+(-1+n)\,\gamma)}$$

0

0

```
(*Writing the profit of a lying firm: type u that announces type d ...*)
FullSimplify[Profit1 /. θ → θu /. q → qd /. {y1 → y1ud, y → yud}]
```

$$\frac{\left(2\,c\,(-2+\gamma)+qd\,\left((-1+n)\,\gamma^2\,(\theta d-\theta u)+4\,\theta u+2\,(-2+n)\,\gamma\,\theta u\right)-2\,(-2+\gamma)\,\mu\right)^2}{4\,(-2+\gamma)^2\,(2+(-1+n)\,\gamma)^2}$$

```
(*Low type when announcing untruthfully*)
y1du =
  y1 /. FullSimplify[Solve[{0 == D[Profit1, y1] /. y → yuu}, {y1}] /. {q → qu, θ → θd}][[1, 1]];
ydu = yuu;
Profit1du = FullSimplify[Profit1 /. {y1 → y1du, y → ydu} /. {θ -> θd , q → qu}] ;
Profitdu = FullSimplify[Profit /. yj → y /. {y1 → y1du, y → ydu} /. {θ -> θd , q → qu} ];
```

```
FullSimplify[y1dd - y1du]
```

$$\text{FullSimplify}\left[\text{y1du} - \left(\text{y1dd} + \left(\frac{+2\,(\text{qu}-\text{qd})\,(2+(-2+n)\,\gamma)\,\theta d + \text{qu}\,(-1+n)\,\gamma^2\,(\theta u - \theta d)}{2\,(2-\gamma)\,(2+(-1+n)\,\gamma)}\right)\right)\right]$$

```
(*This last is a check of the rewriting used in the paper*)
```

$$\frac{-2\,\text{qd}\,(2+(-2+n)\,\gamma)\,\theta d + \text{qu}\,\left(-(-2+\gamma)\,(2+(-1+n)\,\gamma)\,\theta d + (-1+n)\,\gamma^2\,\theta u\right)}{2\,(-2+\gamma)\,(2+(-1+n)\,\gamma)}$$

0

```
(*This is a check of the rewriting used in the paper*)
FullSimplify[ydd - ydu]
```

$$\text{FullSimplify}\left[\text{ydu} - \left(\text{ydd} + \left(\frac{\gamma\,(\text{qd}\,(\theta d - \theta u) - (\text{qu}-\text{qd})\,\theta u)}{(2-\gamma)\,(2+(-1+n)\,\gamma)}\right)\right)\right]$$

$$\text{FullSimplify}\left[\text{ydu} - \left(\text{ydd} + \left(\frac{\gamma\,(\text{qd}\,\theta d - \text{qu}\,\theta u)}{(2-\gamma)\,(2+(-1+n)\,\gamma)}\right)\right)\right]$$

$$\frac{\gamma\,(\text{qd}\,\theta d - \text{qu}\,\theta u)}{(-2+\gamma)\,(2+(-1+n)\,\gamma)}$$

0

0

```
(*The informational Rent is defined as follows...*)
Rent = FullSimplify[Profit1ud - Profit1dd]
FullSimplify[Rent /. θu → θd]
(*This last is a check*)
```

$$\frac{\text{qd}\,(\theta d - \theta u)\,\left(4\,c\,(-2+\gamma) + 4\,\text{qd}\,\theta d + \text{qd}\,\left((-1+n)\,\gamma^2\,(\theta d - \theta u) + 4\,\theta u + 2\,(-2+n)\,\gamma\,(\theta d + \theta u)\right) + 8\,\mu - 4\,\gamma\,\mu\right)}{4\,(-2+\gamma)\,(2+(-1+n)\,\gamma)}$$

0

```
(*The Rent can be conveninently rewritten as follows...*)
```

$$\text{FullSimplify}\left[\frac{\text{qd}\,(\theta d - \theta u)}{4\,(-2+\gamma)\,(2+(-1+n)\,\gamma)}\,\left((2-\gamma)\,4\,(\mu - c) + \text{qd}\,\left((-1+n)\,\gamma^2\,(\theta d - \theta u) + 2\,(2+\,(-2+n)\,\gamma)\,(\theta d + \theta u)\right)\right) - \text{Rent}\right]$$

0

$$\text{FullSimplify}\left[(-1+n)\,\gamma^2\,(\theta d - \theta u) + 2\,(2+\,(-2+n)\,\gamma)\,(\theta d + \theta u)\ /.\ \theta d \to 0\right]$$

```
(*This shows that the Rent is positive since the term qd
  ((-1+n) γ^2 (θd-θu)+2(2+ (-2+n) γ) (θd+θu)) is positive. To see this take the most
  unfavorable cases of θd=0 the expression becomes -(-2+γ) (2+(-1+n) γ) θu>0 *)
```

$$-(-2+\gamma)\,(2+(-1+n)\,\gamma)\,\theta u$$

```
(*Now we derive incentive compatibility*)
IC = FullSimplify[Profit1uu - Profit1du - (Profit1ud - Profit1dd)]
```

$$-\frac{1}{4\,(-2+\gamma)\,(2+(-1+n)\,\gamma)}\,(\theta d-\theta u)\,\Big(4\,c\,(qd-qu)\,(-2+\gamma)+qd^{2}\,\big((4+\gamma\,(-4-\gamma+n\,(2+\gamma)))\,\theta d-(-2+\gamma)\,(2+(-1+n)\,\gamma)\,\theta u\big)-4\,qd\,(-2+\gamma)\,\mu+qu\,\big(qu\,((-2+\gamma)\,(2+(-1+n)\,\gamma)\,\theta d+(-4+\gamma\,(4+\gamma-n\,(2+\gamma)))\,\theta u)+4\,(-2+\gamma)\,\mu\big)\Big)$$

```
(*We now rewrite IC in a compact way...*)

K = 4 (μ - c) (2 - γ);
H = -qd^2 ((2 (2 + (-2 + n) γ) + (-1 + n) γ^2) θd + (2 - γ) (2 + (-1 + n) γ) θu) +
    qu^2 (((2 - γ) (2 + (-1 + n) γ) θd + (2 (2 + (-2 + n) γ) + (-1 + n) γ^2) θu));
FullSimplify[(θu - θd)/(4 (2 - γ) (2 + (-1 + n) γ)) ((qu - qd) K + H)
  - IC]
```

0

```
(*Since K>0 and, whenever qu>qd then H>0,
it follows that monotonicity is sufficient for IC to hold*)
```

Now we move to regulation

```
(*First we verify monotonicity of regulated outputs by
 directly studying the derivative of the info rent w.r.t. qd*)
FullSimplify[D[Rent, qd]]
```

$$\frac{(\theta d-\theta u)\,\Big(2\,c\,(-2+\gamma)+qd\,\big((-1+n)\,\gamma^{2}\,(\theta d-\theta u)+4\,\theta u+2\,(-2+n)\,\gamma\,(\theta d+\theta u)\big)-2\,\gamma\,\mu+4\,(qd\,\theta d+\mu)\Big)}{2\,(-2+\gamma)\,(2+(-1+n)\,\gamma)}$$

```
(*Using the same reasoning we used to show that Rent>0,
we can show that the Rent is increasing in qd*)
(*hence since the Rent enters with a minus in the maximand in the social welfare
 for qd, and since qu instead is undistorted, it follows that indeed qu>qd*)
```

```
(*WELFARE*)
(*regulated demand and regulated consumer surplus*)
pr = mr - q;
CSRu = Integrate[pr, q] /. q → qu;
CSRd = Integrate[pr, q] /. q → qd;

(*unregulated consumers' surplus*)
CSUu = FullSimplify[u /. yj → y /. {y1 → y1uu, y → yuu}]
CSUd = FullSimplify[u /. yj → y /. {y1 → y1dd, y → ydd}]
```

$$-\frac{1}{2(-2+\gamma)^2(2+(-1+n)\gamma)^2}\Big(c^2(-2+\gamma)^2\left(n+2(-1+n)^2\gamma\right)+qu^2(4+\gamma(-8+7\gamma+n(4+(-7+n)\gamma)))\,\theta u^2+2\,qu\,(-2+\gamma)(2+\gamma(-1+(2+(-3+n)n)\gamma))\,\theta u\,\mu-(-2+\gamma)^2(3n+2(-1+n)\gamma)\mu^2-2c(-2+\gamma)\left(qu\left(-2+\gamma\left(3-2n+(-1+n)^2\gamma\right)\right)\theta u+(-2+\gamma)(-n+(2+(-3+n)n)\gamma)\mu\right)\Big)$$

$$-\frac{1}{2(-2+\gamma)^2(2+(-1+n)\gamma)^2}\Big(c^2(-2+\gamma)^2\left(n+2(-1+n)^2\gamma\right)+qd^2(4+\gamma(-8+7\gamma+n(4+(-7+n)\gamma)))\,\theta d^2+2\,qd\,(-2+\gamma)(2+\gamma(-1+(2+(-3+n)n)\gamma))\,\theta d\,\mu-(-2+\gamma)^2(3n+2(-1+n)\gamma)\mu^2-2c(-2+\gamma)\left(qd\left(-2+\gamma\left(3-2n+(-1+n)^2\gamma\right)\right)\theta d+(-2+\gamma)(-n+(2+(-3+n)n)\gamma)\mu\right)\Big)$$

```
Costconglomu = FullSimplify[c (q + y1) - θ q y1 /. q → qu /. θ → θu /. {y1 → y1uu, y → yuu}];
Costconglomd = FullSimplify[c (q + y1) - θ q y1 /. q → qd /. θ → θd /. {y1 → y1dd, y → ydd}];


Costrivalsu = FullSimplify[c y (n - 1) /. {y1 → y1uu, y → yuu}];
Costrivalsd = FullSimplify[c y (n - 1) /. {y1 → y1dd, y → ydd}];


ProfitSep = FullSimplify[Profit1 /. q → 0 /. y1 → y1uu /. y → yuu /. qu → 0]
FullSimplify[Profit /. yj → y /. y1 → y1uu /. y → yuu /. qu → 0]
```

$$\frac{(c-\mu)^2}{(2+(-1+n)\gamma)^2}$$

$$\frac{(c-\mu)^2}{(2+(-1+n)\gamma)^2}$$

```
Welfareu =
  CSRu + CSUu - Costconglomu - Costrivalsu - (1 - a) (n - 1) Profituu - (1 - a) (Rent + ProfitSep);

Welfared = CSRd + CSUd - Costconglomd - Costrivalsd - (1 - a) (n - 1) Profitdd - (1 - a) (ProfitSep);

qusol = qu /. Simplify[Solve[D[Welfareu, qu] == 0, qu]][[1, 1]];

qdsol = qd /. Simplify[Solve[D[Welfared - (1 - a) v / (1 - v) Rent, qd] == 0, qd]][[1, 1]];

FullSimplify[qusol - qdsol /. θd -> θu]
```

0

```
pu = p /. yj → y /. {y1 → y1uu, y → yuu} /. qu → qusol;
p1u = p1 /. yj → y /. {y1 → y1uu, y → yuu} /. qu → qusol;
pd = p /. yj → y /. {y1 → y1dd, y → ydd} /. qd → qdsol;
p1d = p1 /. yj → y /. {y1 → y1dd, y → ydd} /. qd → qdsol;

yu = yuu /. qu → qusol;
y1u = y1uu /. qu → qusol;
yd = ydd /. qd → qdsol;
y1d = y1dd /. qd → qdsol;

ICsol = IC /. {qd → qdsol, qu → qusol};
Rentsol = Rent /. {qd → qdsol, qu → qusol};
```

Now we make some comparative statics on main parameters

```
(*hence it is positive if the numerator is positive which is
 always the case because γ cannot be larger than 1 (see the conditions
   on parameters when defining the preferences u at the very beginning)*)

TestMarginalCostu = c - θu qusol ;
TestMarginalCostd = c - θd qdsol ;
```

```
(*analysis of n*)
parameters = {μ → 10, γ → 0.15, mr → 10, c → 8, θu → 0.2, θd → 0.1, v → 0.5, a → 0.1};

nplotmin = 2;
nplotmax = 10;
ICsolplot = ICsol /. parameters;
TestMarginalCostuplot = TestMarginalCostu /. parameters;
TestMarginalCostdplot = TestMarginalCostd /. parameters;
qusolplot = qusol /. parameters;
qdsolplot = qdsol /. parameters;
p1uplot = p1u /. parameters;
p1dplot = p1d /. parameters;
puplot = pu /. parameters;
pdplot = pd /. parameters;
y1uplot = y1u /. parameters;
y1dplot = y1d /. parameters;
yuplot = yu /. parameters;
ydplot = yd /. parameters;
Yu = y1uplot + (n - 1) yuplot /. parameters;
Yd = y1dplot + (n - 1) ydplot /. parameters;


Rentplot = Rentsol /. parameters;


(*now we can plot...*)

Plot[{ICsolplot, qusolplot - qdsolplot},
 {n, nplotmin, nplotmax}, PlotLabel → {"IC,  qu-qd"}]
Plot[{TestMarginalCostuplot, TestMarginalCostdplot}, {n, nplotmin, nplotmax},
 PlotLabel → {"TestMarginalCostu,  TestMarginalCostd"}]

Plot[{qusolplot, qdsolplot}, {n, nplotmin, nplotmax}, PlotLabel → {"qu,  qd"}]

Plot[{p1uplot, p1dplot}, {n, nplotmin, nplotmax}, PlotLabel → {"p1u,  p1d"}]

Plot[{puplot, pdplot}, {n, nplotmin, nplotmax}, PlotLabel → {"pu,  pd"}]

Plot[{y1uplot, y1dplot}, {n, nplotmin, nplotmax}, PlotLabel → {"y1u,  y1d"}]
Plot[{yuplot, ydplot}, {n, nplotmin, nplotmax}, PlotLabel → {"yu,  yd"}]

Plot[{Yu, Yd}, {n, nplotmin, nplotmax}, PlotLabel → {"Yu,  Yd   (total quantities)"}]

Plot[{Rentplot}, {n, nplotmin, nplotmax}, PlotLabel → {"Rent"}]
```

{IC, qu−qd}



{TestMarginalCostu, TestMarginalCostd}



{qu, qd}



{p1u, p1d}



{pu, pd}



{y1u, y1d}




{yu, yd}
0.90
0.85
0.80
0.75
0.70
0.65
4
6
8
10


{Yu, Yd (total quantities)}
6
5
4
3
4
6
8
10


{Rent}
0.22
0.21
0.20
0.19
0.18
0.17
4
6
8
10

```
(*analysis of γ*)
parameters = {n → 22, μ → 10, mr → 10, c → 8, θu → 0.2, θd → 0.1, v → 0.5, a → 0.1};

ICsolplot = ICsol /. parameters;
TestMarginalCostuplot = TestMarginalCostu /. parameters;
TestMarginalCostdplot = TestMarginalCostd /. parameters;
qusolplot = qusol /. parameters;
qdsolplot = qdsol /. parameters;
p1uplot = p1u /. parameters;
p1dplot = p1d /. parameters;
puplot = pu /. parameters;
pdplot = pd /. parameters;
y1uplot = y1u /. parameters;
y1dplot = y1d /. parameters;
yuplot = yu /. parameters;
ydplot = yd /. parameters;
Yu = y1uplot + (n - 1) yuplot /. parameters;
Yd = y1dplot + (n - 1) ydplot /. parameters;


Rentplot = Rentsol /. parameters;


(*now we can plot...*)

Plot[{ICsolplot, qusolplot - qdsolplot}, {γ, 0, 0.99}, PlotLabel → {"IC,  qu-qd"}]
Plot[{TestMarginalCostuplot, TestMarginalCostdplot},
 {γ, 0, 0.99}, PlotLabel → {"TestMarginalCostu,  TestMarginalCostd"}]

Plot[{qusolplot, qdsolplot}, {γ, 0, 0.99}, PlotLabel → {"qu,  qd"}]

Plot[{p1uplot, p1dplot}, {γ, 0, 0.99}, PlotLabel → {"p1u,  p1d"}]

Plot[{puplot, pdplot}, {γ, 0, 0.99}, PlotLabel → {"pu,  pd"}]

Plot[{y1uplot, y1dplot}, {γ, 0, 0.99}, PlotLabel → {"y1u,  y1d"}]
Plot[{yuplot, ydplot}, {γ, 0, 0.99}, PlotLabel → {"yu,  yd"}]

Plot[{Yu, Yd}, {γ, 0, 0.99}, PlotLabel → {"Yu,  Yd   (total quantities)"}]

Plot[{Rentplot}, {γ, 0, 0.99}, PlotLabel → {"Rent"}]
```



{TestMarginalCostu, TestMarginalCostd}



{qu, qd}



{p1u, p1d}



{pu, pd}




{y1u, y1d}
0.9
0.8
0.7
0.6
0.5
0.4
0.2
0.4
0.6
0.8
1.0


{yu, yd}
0.7
0.6
0.5
0.4
0.3
0.2
0.2
0.4
0.6
0.8
1.0


{Yu, Yd (total quantities)}
14
12
10
8
6
4
0.2
0.4
0.6
0.8
1.0


{Rent}
0.14
0.13
0.12
0.11
0.10
0.09
0.08
0.2
0.4
0.6
0.8
1.0

```
(*analysis of θu*)
parameters = {n → 22, γ → 0.15, μ → 10, mr → 10, c → 8, θd → 0.1, v → 0.5, a → 0.1};

ICsolplot = ICsol /. parameters;
TestMarginalCostuplot = TestMarginalCostu /. parameters;
TestMarginalCostdplot = TestMarginalCostd /. parameters;
qusolplot = qusol /. parameters;
qdsolplot = qdsol /. parameters;
p1uplot = p1u /. parameters;
p1dplot = p1d /. parameters;
puplot = pu /. parameters;
pdplot = pd /. parameters;
y1uplot = y1u /. parameters;
y1dplot = y1d /. parameters;
yuplot = yu /. parameters;
ydplot = yd /. parameters;
Yu = y1uplot + (n - 1) yuplot /. parameters;
Yd = y1dplot + (n - 1) ydplot /. parameters;


Rentplot = Rentsol /. parameters;


(*now we can plot...*)

Plot[{ICsolplot, qusolplot - qdsolplot}, {θu, 0.1, 0.5}, PlotLabel → {"IC,  qu-qd"}]
Plot[{TestMarginalCostuplot, TestMarginalCostdplot},
 {n, nplotmin, nplotmax}, PlotLabel → {"TestMarginalCostu,  TestMarginalCostd"}]

Plot[{qusolplot, qdsolplot}, {θu, 0.1, 0.5}, PlotLabel → {"qu,  qd"}]

Plot[{p1uplot, p1dplot}, {θu, 0.1, 0.5}, PlotLabel → {"p1u,  p1d"}]

Plot[{puplot, pdplot}, {θu, 0.1, 0.5}, PlotLabel → {"pu,  pd"}]

Plot[{y1uplot, y1dplot}, {θu, 0.1, 0.5}, PlotLabel → {"y1u,  y1d"}]
Plot[{yuplot, ydplot}, {θu, 0.1, 0.5}, PlotLabel → {"yu,  yd"}]

Plot[{Yu, Yd}, {θu, 0.1, 0.5}, PlotLabel → {"Yu,  Yd   (total quantities)"}]

Plot[{Rentplot}, {θu, 0.1, 0.5}, PlotLabel → {"Rent"}]
```



{TestMarginalCostu, TestMarginalCostd}



{qu, qd}



{p1u, p1d}



{pu, pd}



{y1u, y1d}



{yu, yd}



{Yu, Yd (total quantities)}



{Rent}



```
(*analysis of v*)
parameters = {n → 22, γ → 0.15, μ → 10, mr → 10, c → 8, θu → 0.2, θd → 0.1, a → 0.1};

ICsolplot = ICsol /. parameters;
TestMarginalCostuplot = TestMarginalCostu /. parameters;
TestMarginalCostdplot = TestMarginalCostd /. parameters;
qusolplot = qusol /. parameters;
qdsolplot = qdsol /. parameters;
p1uplot = p1u /. parameters;
p1dplot = p1d /. parameters;
puplot = pu /. parameters;
pdplot = pd /. parameters;
y1uplot = y1u /. parameters;
y1dplot = y1d /. parameters;
yuplot = yu /. parameters;
ydplot = yd /. parameters;
Yu = y1uplot + (n - 1) yuplot /. parameters;
Yd = y1dplot + (n - 1) ydplot /. parameters;


Rentplot = Rentsol /. parameters;


(*now we can plot...*)

Plot[{ICsolplot, qusolplot - qdsolplot}, {v, 0, 1}, PlotLabel → {"IC,  qu-qd"}]
Plot[{TestMarginalCostuplot, TestMarginalCostdplot},
 {n, nplotmin, nplotmax}, PlotLabel → {"TestMarginalCostu,  TestMarginalCostd"}]

Plot[{qusolplot, qdsolplot}, {v, 0, 1}, PlotLabel → {"qu,  qd"}]

Plot[{p1uplot, p1dplot}, {v, 0, 1}, PlotLabel → {"p1u,  p1d"}]

Plot[{puplot, pdplot}, {v, 0, 1}, PlotLabel → {"pu,  pd"}]

Plot[{y1uplot, y1dplot}, {v, 0, 1}, PlotLabel → {"y1u,  y1d"}]
Plot[{yuplot, ydplot}, {v, 0, 1}, PlotLabel → {"yu,  yd"}]

Plot[{Yu, Yd}, {v, 0, 1}, PlotLabel → {"Yu,  Yd   (total quantities)"}]

Plot[{Rentplot}, {v, 0, 1}, PlotLabel → {"Rent"}]
```




{TestMarginalCostu, TestMarginalCostd}
14
12
10
8
6
4
2
4
6
8
10


{qu, qd}
2.3
2.2
2.1
2.0
1.9
1.8
1.7
0.2
0.4
0.6
0.8
1.0


{p1u, p1d}
8.50
8.45
8.40
8.35
8.30
8.25
0.2
0.4
0.6
0.8
1.0


{pu, pd}
8.392
8.390
8.388
8.386
8.384
0.2
0.4
0.6
0.8
1.0

{y1u, y1d}



{yu, yd}



{Yu, Yd (total quantities)}



{Rent}